

```

proc iml;

friction = 2.5; *Friction coefficient when traveling from a model tract to
Coltsville NHP;

Basic = 26;
EastArmory = 57;
FullSite = 179;

impact = FullSite; *Expected number of jobs (26, 57, or 179 under the Basic,
East Armory, and Full Site scenario, respectively);

use sasdata.census_blocks; * Block groups data set;
read all var _ALL_ into zones[colname = varNames]; * Read all contents of
data set into matrix called zones;
close sasdata.census_blocks;

minutes = zones[, 5];
step1 = minutes ## friction; *Raise minutes, elementwise, to the power of the
friction coefficient;
step2 = step1 ## -1; *Find the reciprocal of step 1;
step3 = sum(step2);
step4 = step2 / step3; *Scaling factor;
attraction = zones[, 6]; *attraction = attractiveness of a zone to
development;
step5 = step4 # attraction; *Multiply each zone's accessibility by that
zone's attraction;
step6 = step5 * impact; *Total impact (unconstrained);
step7 = sum(step6);
step8 = step6 / step7; *This and above: find each zone's share of total
impact;
step9 = step8 * impact;

tractID = zones[, 2];

names = {Jobs, Tract};
result = j(nrow(zones), 2, 0);

result[, 1] = step9;
result[, 2] = tractID;

create sasdata.C_Block_Max_25 from result[colname = names]; *modify output
name here;
append from result;
close sasdata.C_Block_Max_25; *modify output name here;
quit;

proc summary data = sasdata.C_Block_Max_25 sum print; *modify output name
here;
var Jobs; * checks to see if the script works properly by ensuring that the
number of jobs allocated matches the model's input;
run;

*export the output to an Excel spreadsheet;
proc export
data = sasdata.C_Block_Max_25
dbms = xlsx

```

```
outfile = "P:\Statistics\FromSAS\C_Block_Max_25.xlsx"  
replace;  
run;
```